```
(* Evaluate each cell by pressing shift-return or the enter key on the numeric keypad *)

(* Insert matrices by right clicking and
 selecting insert table/matrix and then selecting matrix *)
```

$$a = \begin{pmatrix} 1 & 2 \\ 2 & 4 \end{pmatrix};$$

```
(* You can also enter matrices as a list of lists with just the keyboard as follows *)

b = {{1, 2}, {3, 4}};

(* We can display b as a matrix by typing //MatrixForm *)

b // MatrixForm
```

$$\begin{pmatrix} 1 & 2 \\ 3 & 4 \end{pmatrix}$$

```
(* Bigger matrices are fine too *)
```

$$c = \begin{pmatrix} 1 & 2 & 3 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix};$$

```
(* You can enter vectors the same way *)
```

$$v = \begin{pmatrix} 1 \\ 2 \end{pmatrix};$$

```
(* Vectors and matrices can have variables *)
```

$$d = \begin{pmatrix} q & w \\ r & t \end{pmatrix};$$

$$v_2 = \begin{pmatrix} x \\ y \\ z \end{pmatrix};$$

```
(* Find the norm of a vector with norm *)

Norm[v]
```

$$\sqrt{5}$$

```
(* The norm formula is slightly more complicated if your scalars
 can be complex instead of real, but that won't affect you for 568 *)
```

$\text{Norm}[v_2]$

$$\sqrt{x\,\text{Conjugate}[x] + y\,\text{Conjugate}[y] + z\,\text{Conjugate}[z]}$$

```
(* Take the projection of v=(x,y)
  onto the line with direction vector (D1,D2) *)
```

$\text{Projection}[\{x, y\}, \{D_1, D_2\}, \text{Dot}] \text{ // MatrixForm}$

$$\begin{pmatrix} \frac{D_1 (x D_1 + y D_2)}{D_1^2 + D_2^2} \\ \frac{D_2 (x D_1 + y D_2)}{D_1^2 + D_2^2} \end{pmatrix}$$

```
(* Multiply matrices using . *)
```

```
a.v // MatrixForm
```

$$\begin{pmatrix} 5 \\ 10 \end{pmatrix}$$

```
d.v // MatrixForm
```

$$\begin{pmatrix} q + 2\,w \\ r + 2\,t \end{pmatrix}$$

```
d.d // MatrixForm
```

$$\begin{pmatrix} q^2 + r\,w & q\,w + t\,w \\ q\,r + r\,t & t^2 + r\,w \end{pmatrix}$$

```
(* This is the same as raising d to the second power, which we can do as follows: *)

MatrixPower[d, 2] // MatrixForm
```

$$\begin{pmatrix} q^2 + r\,w & q\,w + t\,w \\ q\,r + r\,t & t^2 + r\,w \end{pmatrix}$$

```
(* If there is a general form for a matrix to a power,
can solve for that as well.  Here we raise a matrix to the nth power *)
```

MatrixPower$\left[\begin{pmatrix} 1 & 1 \\ 0 & 1 \end{pmatrix}, n\right]$ // MatrixForm

$$\begin{pmatrix} 1 & n \\ 0 & 1 \end{pmatrix}$$

```
(* The inverse is the same as taking the matrix to the -1 power,
but the shorter way is to just use Inverse *)
MatrixPower[d, -1] // MatrixForm
```

$$\begin{pmatrix} \frac{t}{q\,t - r\,w} & -\frac{w}{q\,t - r\,w} \\ -\frac{r}{q\,t - r\,w} & \frac{q}{q\,t - r\,w} \end{pmatrix}$$

```
Inverse[d] // MatrixForm
```

$$\begin{pmatrix} \frac{t}{q\,t - r\,w} & -\frac{w}{q\,t - r\,w} \\ -\frac{r}{q\,t - r\,w} & \frac{q}{q\,t - r\,w} \end{pmatrix}$$

```
(* Many matrices are not invertible, but every matrix has a pseudo
 inverse.  You use the pseudo inverse when finding least squares solutions. *)
```

PseudoInverse$\left[\begin{pmatrix} 1 & 2 \\ 1 & 2.1 \\ 2 & 3 \end{pmatrix}\right]$ // MatrixForm

$$\begin{pmatrix} -1.13878 & -1.55102 & 1.8449 \\ 0.77551 & 1.02041 & -0.897959 \end{pmatrix}$$

```
(* Take the transpose of a matrix with Transpose *)
```

```
Transpose[d] // MatrixForm
```

$$\begin{pmatrix} q & r \\ w & t \end{pmatrix}$$

```
(* Recall:  A matrix is symmetric if it equals its own transpose,
so let's test to see if a and b are symmetric *)

a == Transpose[a]
```

True

```
b == Transpose[b]
```

False

```
(* That is, a is symmetric and b is not.  Note we use two =
 to test for equality.  One equal sign is an assignment *)

(* We can row reduce matrices or find their nullspace *)

(* NullSpace returns a basis for the null space,
but transposed from the way you are used to, so we transpose it *)

Transpose[NullSpace[c]] // MatrixForm
```

$$\begin{pmatrix} -3 & -2 \\ 0 & 1 \\ 1 & 0 \end{pmatrix}$$

```
(* That is, the vectors (-3;0;1) and (-2;1;0) form a basis for the nullspace of c *)

(* Since b is invertible, the only vector in the null space is the zero vector...
 the nullspace is zero dimensional (just one point), so the basis set is empty *)

NullSpace[b] // MatrixForm
```

{}

```
(* Row Reducing *)

RowReduce[a] // MatrixForm
```

$$\begin{pmatrix} 1 & 2 \\ 0 & 0 \end{pmatrix}$$

```
RowReduce[b] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

```
(* Recall we can use row reducing to solve an augmented matrix.  For example: *)
```

```
RowReduce[(1 2 1
           4 5 2)] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & -\frac{1}{3} \\ 0 & 1 & \frac{2}{3} \end{pmatrix}$$

```
(* That is, we have solved the linear system of equations:
   x + 2y == 1 and
     4x + 5 y == 2
    for x and y, and found x=-1/3 and y=2/3 *)

(* Alternatively, we could have just used the solve command *)

Solve[{x + 2 y == 1, 4 x + 5 y == 2}, {x, y}]
```

$\{\{x \to -\frac{1}{3}, y \to \frac{2}{3}\}\}$

```
(* Again, symbolic expressions are fine... suppose the right hand sides were q and w *)

Solve[{x + 2 y == q, 4 x + 5 y == w}, {x, y}]
```

$\{\{x \to \frac{1}{3}(-5 q + 2 w), y \to \frac{1}{3}(4 q - w)\}\}$

```
(* We can find the eigenvalues and eigenvectors of matrices *)

(* There's multiple ways to do this *)

(* First:  From the definition.  We recall the
  eigenvalues of b are the roots of Det[b-λ IdentityMatrix[2]] *)

Det[b - λ IdentityMatrix[2]]
```

$-2 - 5 \lambda + \lambda^2$

```
(* Alternatively,
this polynomial is known as the characteristic polynomial of the matrix *)

CharacteristicPolynomial[b, λ]
```

$-2 - 5 \lambda + \lambda^2$

```
(* We can then find when the characteristic polynomial is zero *)

(* Some polynomials factor nicely. *)

Factor[λ^2 - 3 λ + 2]
```

$(-2 + \lambda)\ (-1 + \lambda)$

```
(* Ours, however, does not *)

Factor[-2 - 5 λ + λ^2]
```

$-2 - 5 \lambda + \lambda^2$

```
(* That's okay though.  We can solve for when the characteristic polynomial equals zero. *)
```

```
Solve[-2 - 5 λ + λ^2 == 0, λ]
```

$\left\{\left\{\lambda \to \frac{1}{2}\left(5-\sqrt{33}\right)\right\}, \left\{\lambda \to \frac{1}{2}\left(5+\sqrt{33}\right)\right\}\right\}$

(* Thus the eigenvalues of b are $\frac{1}{2}\left(5\pm\sqrt{33}\right)$ *)

```
(* If you only want to know the eigenvalues, just use: *)

Eigenvalues[b]
```

$\left\{\frac{1}{2}\left(5+\sqrt{33}\right), \frac{1}{2}\left(5-\sqrt{33}\right)\right\}$

(* To find just the eigenvectors,
use: (again, due to the way mathematica works, you probably want to take the transpose.)
Here the eigenvectors are $\begin{pmatrix} -\frac{4}{3}+\frac{1}{6}\left(5+\sqrt{33}\right) \\ 1 \end{pmatrix}$ and $\begin{pmatrix} -\frac{4}{3}+\frac{1}{6}\left(5-\sqrt{33}\right) \\ 1 \end{pmatrix}$ *)

```
Transpose[Eigenvectors[b]] // MatrixForm
```

$$\begin{pmatrix} -\frac{4}{3}+\frac{1}{6}\left(5+\sqrt{33}\right) & -\frac{4}{3}+\frac{1}{6}\left(5-\sqrt{33}\right) \\ 1 & 1 \end{pmatrix}$$

```
(* We can check by multiplying the matrix by the
 vector and comparing it to the eigenvalue times the vector *)
```

$$b . \begin{pmatrix} -\frac{4}{3}+\frac{1}{6}\left(5+\sqrt{33}\right) \\ 1 \end{pmatrix} == \frac{1}{2}\left(5+\sqrt{33}\right) \begin{pmatrix} -\frac{4}{3}+\frac{1}{6}\left(5+\sqrt{33}\right) \\ 1 \end{pmatrix}$$

```
True

(* Alternatively,
use Eigensystem to find both eigenvalues and eigenvectors at the same time *)

{evals, evecs} = Eigensystem[b];

(* The 2nd eigenvalue is: *)

evals[[2]]
```

$$\frac{1}{2}\left(5-\sqrt{33}\right)$$

```
(* and it has corresponding eigenvector *)

evecs[[2]] // MatrixForm
```

$$\begin{pmatrix} -\frac{4}{3}+\frac{1}{6}\left(5-\sqrt{33}\right) \\ 1 \end{pmatrix}$$

```
(* Remember the determinant of a matrix is the product of the eigenvalues *)

Det[b]

-2
```

```
evals[[1]] * evals[[2]]
```

$$\frac{1}{4}\left(5-\sqrt{33}\right)\left(5+\sqrt{33}\right)$$

```
(* They don't look the same, do they?  We're going to have to simplify. *)

FullSimplify[evals[[1]] * evals[[2]]]
```

-2

```
(* This is the same as the determinant *)
```

(* The trace of b, the sum of the diagonal entries, is $b_{11}+b_{22}$ *)

```
Tr[b]
```

5

```
b[[1, 1]] + b[[2, 2]]
```

5

```
(* The trace is also the sum of the eigenvalues *)

FullSimplify[evals[[1]] + evals[[2]]]
```

5

```
(* To do Gram-Schmidt, use orthogonalize *)

Orthogonalize[{{1, 2, 3, 0}, {4, 2, 1, 0}, {3, 1, 2, 1}}]
```

$$\left\{\left\{\frac{1}{\sqrt{14}}, \sqrt{\frac{2}{7}}, \frac{3}{\sqrt{14}}, 0\right\}, \left\{\frac{45}{\sqrt{2422}}, 3\sqrt{\frac{2}{1211}}, -\frac{19}{\sqrt{2422}}, 0\right\}, \left\{\frac{26\sqrt{\frac{2}{3287}}}{3}, -\frac{143}{3\sqrt{6574}}, 13\sqrt{\frac{2}{3287}}, \frac{\sqrt{\frac{173}{38}}}{3}\right\}\right\}$$

(* That is, the vectors $\left(\frac{1}{\sqrt{14}},\sqrt{\frac{2}{7}},\frac{3}{\sqrt{14}},0\right)$, $\left(\frac{45}{\sqrt{2422}},3\sqrt{\frac{2}{1211}},-\frac{19}{\sqrt{2422}},0\right)$,
etc... are orthogonal vectors, i.e. the dotproduct of any two is zero,
with the same span as the original vectors (1,2,3,0), (4,2,1,0) and (3,1,2,1). *)

```
(* QR decomposition of b gives a matrix Q with orthogonal
  columns and an upper triangular matrix R such that Q.R=b.  As before,
we need to transpose the Q mathematica gives. *)

{Q, R} = QRDecomposition[b];  Q = Transpose[Q];

Q // MatrixForm
```

$$\begin{pmatrix} \frac{1}{\sqrt{10}} & \frac{3}{\sqrt{10}} \\ \frac{3}{\sqrt{10}} & -\frac{1}{\sqrt{10}} \end{pmatrix}$$

```
R // MatrixForm
```

$$\begin{pmatrix} \sqrt{10} & 7\sqrt{\frac{2}{5}} \\ 0 & \sqrt{\frac{2}{5}} \end{pmatrix}$$

```
Q.R // MatrixForm
```

$$\begin{pmatrix} 1 & 2 \\ 3 & 4 \end{pmatrix}$$

```
(* Q is an orthogonal matrix if and only if Transpose[Q]==Inverse[Q].
   Let's check that Q is orthogonal. *)

Transpose[Q] == Inverse[Q]
```

True

```
(* Recall that the determinant of an orthogonal matrix is
 ±1.  Be careful: the converse is false. *)

Det[Q]
```

-1

```
(* For any nonsingular matrix A, there exists matrices P, L and U such that P.A=L.U,
where P is a permutation matrix (think row interchanges),
L is lower triangular with ones on the diagonal and U is upper
 triangular.  Mathematica returns the lower part of L in the same matrix as U. *)

{mLU, pivots, conditionNum} = LUDecomposition[b]; mLU = Transpose[mLU];

mLU // MatrixForm
```

$$\begin{pmatrix} 1 & 3 \\ 2 & -2 \end{pmatrix}$$

(* That is, L= $\begin{pmatrix} 1 & 0 \\ 2 & 1 \end{pmatrix}$ and U= $\begin{pmatrix} 1 & 3 \\ 0 & -2 \end{pmatrix}$. Let's check. *)

$\begin{pmatrix} 1 & 0 \\ 2 & 1 \end{pmatrix} . \begin{pmatrix} 1 & 3 \\ 0 & -2 \end{pmatrix}$ // MatrixForm

$$\begin{pmatrix} 1 & 3 \\ 2 & 4 \end{pmatrix}$$

```
(* So L.U is b, up to permutation of rows. *)

(* The singular value decomposition decomposes an m x n matrix A into
  A=U.D.Transpose[V],
where U and V are orthogonal (unitary in the complex case) and
 d is diagonal.  The diagonal entries of d are called the singular values of A. *)

{u, d, v} = SingularValueDecomposition[b];
```

```
FullSimplify[u] // MatrixForm
```

$$\begin{pmatrix} \sqrt{\frac{1}{2} - \frac{5}{\sqrt{221}}} & \sqrt{\frac{1}{2} + \frac{5}{\sqrt{221}}} \\ \sqrt{\frac{1}{2} + \frac{5}{\sqrt{221}}} & -\sqrt{\frac{1}{2} - \frac{5}{\sqrt{221}}} \end{pmatrix}$$

```
FullSimplify[d] // MatrixForm
```

$$\begin{pmatrix} \sqrt{15 + \sqrt{221}} & 0 \\ 0 & \sqrt{15 - \sqrt{221}} \end{pmatrix}$$

```
FullSimplify[v] // MatrixForm
```

$$\begin{pmatrix} \sqrt{\frac{1}{2} - \frac{5}{2\sqrt{221}}} & -\sqrt{\frac{1}{2} + \frac{5}{2\sqrt{221}}} \\ \sqrt{\frac{1}{2} + \frac{5}{2\sqrt{221}}} & \sqrt{\frac{1}{2} - \frac{5}{2\sqrt{221}}} \end{pmatrix}$$

```
(* The product u.d.Transpose[v] is the original matrix *)

FullSimplify[u.d.Transpose[v]] // MatrixForm
```

$$\begin{pmatrix} 1 & 2 \\ 3 & 4 \end{pmatrix}$$